

INSTABILIDADE DE ELEMENTOS ESTRUTURAIS MULTIESTÁVEIS

Aluna: Ayla Pedron Hajjar

Orientador: Paulo Batista Gonçalves

2020

Introdução:

Atualmente as estruturas vêm se tornando cada vez mais leves e esbeltas, garantindo economia e otimização dos elementos estruturais. Esta modernização nas concepções de projetos pode ser observada nos mais diversos campos das engenharias civil, marítima, mecânica e aeroespacial, tornando-se alvo de estudos ao redor do mundo. Não só isso, o desenvolvimento da pesquisa sobre o tema se mostra cada vez mais importante em razão da estabilidade e segurança das estruturas, uma vez que as não linearidades estruturais se tornam mais frequentes, sendo as estruturas influenciadas negativamente por carregamentos estáticos, dinâmicos e vibrações excessivas.

Neste âmbito, têm sido desenvolvidas teorias acerca da estabilidade estática e dinâmica, e de sistemas dinâmicos não lineares, se debruçando no estudo das bifurcações geradas pelas variações nos parâmetros de controle. Desta forma, este trabalho pretende apresentar uma análise inicial acerca do efeito da não linearidade geométrica em estruturas multiestáveis.

Objetivos:

O objetivo é apresentar o cálculo da energia potencial total de sistemas estruturais, exibindo suas equações de equilíbrio e posições de estabilidade e instabilidade, proporcionando o entendimento dos fenômenos de instabilidade para desenvolvimento de prescrições e normas de projeto, contribuindo para um projeto mais econômico e seguro.

Metodologia:

O conteúdo foi desenvolvido com o auxílio do software MAPLE, levando em consideração a seguinte metodologia:

1. Escolha dos modelos de estruturas a serem analisados.
2. Desenvolvimento das equações analíticas considerando a não linearidade geométrica, utilizando o Princípio da Energia Potencial Mínima.
3. Análise do comportamento estrutural não linear com a identificação dos caminhos não lineares de equilíbrio e seus pontos de bifurcação em função dos parâmetros físicos e geométricos do sistema.
4. Relatórios de pesquisa e trabalhos para publicação.

Apresentação do trabalho:

O presente trabalho foi desenvolvido em três partes. Inicialmente, a partir da compreensão teórica acerca do tema estabilidade de estruturas, foi possível o entendimento de fatores que afetam esse equilíbrio. Em seguida, utilizando o Software MAPLE para aplicar os critérios de energia potencial mínima e calcular as cargas críticas das estruturas, foram analisados modelos estruturais de forma a compreender os quatro tipos básicos de bifurcação: simétrica estável, simétrica instável, assimétrica e bifurcação por ponte limite. As representações gráficas dos sistemas foram realizadas com auxílio do Software Ftool. Finalmente, também foram estudadas as consequências das variações dos parâmetros iniciais de posição, analisando as mudanças provocadas nas configurações de equilíbrio dessas estruturas imperfeitas.

Formulação:

Um sistema elástico sob ação de forças conservativas possui um potencial gravitacional (V), que é matematicamente simétrico ao trabalho das forças externas atuantes (W_e), e uma energia interna de deformação (U). A variação da energia potencial total (π ou var_pot) de tais sistemas, pode ser definida pela seguinte equação:

$$var_pot(\theta) = U(\theta) + V(\theta) = U(\theta) - W_e(\theta)$$

O Princípio da Energia Potencial Mínima indica que, para que um sistema elástico sob forças conservativas esteja em equilíbrio mecânico, este deve apresentar energia potencial total estacionária [1]. Sendo assim, para determinar sua configuração de equilíbrio, tem-se:

$$\frac{d}{d\theta} var_pot(\theta) = 0$$

Deste modo, pretende-se formular, a partir do princípio citado, as equações não lineares de equilíbrio de diferentes sistemas elásticos, deduzir suas cargas críticas de projeto, e identificar as configurações de estabilidade e instabilidade, a partir do estudo de derivadas de ordens superiores. Ainda, analisando as mesmas estruturas, porém com imperfeições iniciais, determinar a consequência dessas imperfeições nos caminhos de equilíbrio das barras.

1. Barra rígida com mola de torção e carga transversal ao eixo




Figura 1 – Posições Inicial e Final

No MAPLE:

> **U:=theta->1/2*k*theta^2;**

$$U := \theta \rightarrow \frac{1}{2} k \theta^2$$

> **We:=theta->P*L*sin(theta);**

$$We := \theta \rightarrow P L \sin(\theta)$$

> **var_pot:=theta-> U(theta)-We(theta);**

$$var_pot := \theta \rightarrow U(\theta) - We(\theta)$$

> **var_pot(theta);**

$$\frac{k \theta^2}{2} - P L \sin(\theta)$$

A variação de energia potencial total é estacionária no equilíbrio, logo, quando sua derivada é nula, tem-se a posição de equilíbrio do conjunto. Resolvendo-a para P, tem-se ainda o caminho não linear de equilíbrio. A Figura 2 mostra o caminho não linear de equilíbrio. Linearizando a equação resultante tem-se uma aproximação linear da resposta da estrutura que é válida para pequenos deslocamentos. A partir de $\theta = 0.6$ já começa a existir uma diferença entre as duas soluções, evidenciando a importância da não linearidade geométrica. A rigidez efetiva K_{ef} (tangente à curva) mostra ser este um sistema estrutural com ganho de rigidez.

> **solve(D(var_pot)(theta)=0,P);**

$$\frac{k \theta}{L \cos(\theta)}$$

> **lambda:=theta->theta/cos(theta); (lambda = PL/k)**

$$\lambda := \theta \rightarrow \frac{\theta}{\cos(\theta)}$$

> **plot([lambda(theta),theta],theta=0..1.5,lambda=0..20);**




Figura 2 – Caminhos de Equilíbrio Não Linear e Linear: Estrutura com ganho de rigidez.

> **P:=theta->k*theta/(L*cos(theta));**

$$P := \theta \rightarrow \frac{k \theta}{L \cos(\theta)}$$

> **simplify(D(P)(theta));** Rigidez efetiva: K_{ef}

$$\frac{k (\sin(\theta) \theta + \cos(\theta))}{L \cos(\theta)^2}$$

2. Barra rígida com mola lateral de torção




Figura 3 – Posições Inicial e Final

No MAPLE:

> **U:=theta->1/2*k*theta^2;**

$$U := \theta \rightarrow \frac{1}{2} k \theta^2$$

> **We:=theta->P*L*(1-cos(theta));**

$$W_e := \theta \rightarrow PL(1 - \cos(\theta))$$

> **var_pot:=theta-> U(theta)-We(theta);**

$$\text{var_pot} := \theta \rightarrow U(\theta) - W_e(\theta)$$

> **var_pot(theta);**

$$\frac{k \theta^2}{2} - PL(1 - \cos(\theta))$$

> **D(var_pot)(theta)=0;**

$$k \theta - PL \sin(\theta) = 0$$

1ª Posição de Equilíbrio: (barra não deformada) - caminho fundamental

> **theta=0;**

$$\theta = 0$$

2ª Posição de Equilíbrio: (barra deslocada devido a carga P) - caminho pós crítico

> **solve(D(var_pot)(theta)=0,P);**

$$\frac{k \theta}{L \sin(\theta)}$$

Na 2ª Posição, para ângulos muito pequenos:

> **solve([P=k*theta/L/sin(theta),sin(theta)=theta]);** (carga crítica $P_{cr} = k/L$)

$$\left\{ L = \frac{k}{P}, P = P, k = k, \theta = 0 \right\}$$

Para avaliar a estabilidade dos caminhos, deve-se estudar a segunda derivada da energia potencial do sistema.

> **D(D(var_pot))(theta);**

$$k - PL \cos(\theta)$$

Caminho Fundamental:

> **solve([D(D(var_pot))(theta)=0,cos(theta)=1]);**

$$\left\{ L = L, P = P, k = LP, \theta = 0 \right\}$$

$$\text{Logo, } P_{cr} = \frac{k}{L}.$$

> **solve([D(D(var_pot))(theta)>0,L*cos(theta)>0],P);**

$$\left\{ \left[\left\{ P < \frac{k}{L \cos(\theta)} \right\} \right] \quad 0 < L \cos(\theta) \right. \\ \left. \left[\quad \right] \quad \text{otherwise} \right.$$

Segunda derivada apresenta valores positivos quando $P < P_{cr}$. Assim, quando $P < P_{cr}$, tem-se equilíbrio estável.

> **solve([D(D(var_pot))(theta)<0,L*cos(theta)>0],P);**

$$\left\{ \begin{array}{ll} \left[\left\{ \frac{k}{L \cos(\theta)} < P \right\} \right] & 0 < L \cos(\theta) \\ [] & otherwise \end{array} \right.$$

De mesmo modo, quando $P > P_{cr}$, tem-se equilíbrio instável

> **solve([D(D(var_pot))(theta)=0,cos(theta)=1]);**

$$\{ L = L, P = P, k = L P, \theta = 0 \}$$

Quando a segunda derivada é igual a zero, $P = P_{cr}$, o que não é suficiente para determinar a estabilidade do caminho, deve-se analisar as derivadas de ordem superior.¹

Caminho Pós-Crítico:

> **subs(P=k*theta/L/sin(theta),D(D(var_pot))(theta));**

$$k - \frac{k \theta \cos(\theta)}{\sin(\theta)}$$

> **solve([k-k*theta/sin(theta)*cos(theta)>0,k>0]);**

$$\{ \theta = \theta, 0 < k \}$$

A segunda derivada é positiva para todo valor de θ , logo o caminho pós crítico é sempre estável.

Expansão de Taylor de $\text{var_pot}(\theta=0)$:

> **taylor(1/2*k*theta^2-P*L*(1-cos(theta)),theta=0);**

$$\left(\frac{k}{2} - \frac{L P}{2} \right) \theta^2 + \frac{L P}{24} \theta^4 + O(\theta^6)$$

Sendo $k=P*L > 0$, o primeiro termo é nulo, mas o próximo $(PL/24)$ é *positivo*, sendo assim, em $P = P_{cr}$ para $\theta = 0$, a configuração é estável.

> **plot([1/2*theta^2-1/2*(1-cos(theta)), 1/2*theta^2-(1-cos(theta)), 1/2*theta^2-2*(1-cos(theta))], theta=-3..3, var_pot=-1..1, color=[red,green,blue]);**

¹ Para analisar derivadas de ordem superior, podemos fazer a expansão em série de Taylor de var_pot em $\theta = 0$, e determinar a primeira derivada não nula (porque enquanto $d(\text{var_pot})/d\theta = 0$, não podemos concluir nada; na primeira não nula, estudamos quando é positiva ou negativa, para então concluirmos sobre a estabilidade deste ponto de bifurcação).




Figura 4 – Variação da Energia Potencial

Até $P = P_{cr}$, a estrutura apresenta apenas um mínimo (Figura 4), sendo este equilíbrio estável. A partir de $P > P_{cr}$, o gráfico da variação de energia potencial apresenta dois vales, representando duas possíveis configurações de equilíbrio estável, e um máximo, na origem, que representa um equilíbrio instável da configuração fundamental de equilíbrio.

Como analogia a uma situação física, nos gráficos $P < P_{cr}$ e $P = P_{cr}$, um objeto introduzido ali sempre se manteria na mesma posição. Porém, ao aumentar a carga, de modo que fique maior que a carga crítica ($P > P_{cr}$, gráfico azul), o objeto não conseguiria se manter estável, sendo obrigado descer para um dos vales em busca da estabilidade. Uma estrutura como um pilar, por exemplo, se manteria na vertical até $P = P_{cr}$, mas depois, se $P > P_{cr}$, passaria a sofrer algumas deformações de flexão e possíveis danos.

3. Barra rígida com mola lateral de torção com imperfeição inicial β




Figura 5 – Posições Inicial e Final

No MAPLE:

> **U:=theta->1/2*k*(theta-beta)^2;**

$$U := \theta \rightarrow \frac{1}{2} k (\theta - \beta)^2$$

> **We:=theta->P*L*(cos(beta)-cos(theta));**

$$We := \theta \rightarrow P L (\cos(\beta) - \cos(\theta))$$

> **var_pot:=theta->U(theta)-We(theta);**

$$var_pot := \theta \rightarrow U(\theta) - We(\theta)$$

> **var_pot(theta);**

$$\frac{k (\theta - \beta)^2}{2} - P L (\cos(\beta) - \cos(\theta))$$

> **D(var_pot)(theta);**

$$k (\theta - \beta) - P L \sin(\theta)$$

> solve(D(var_pot)(theta)=0,P);

$$-\frac{k(\beta - \theta)}{L \sin(\theta)}$$

> D(D(var_pot))(theta);

$$k - PL \cos(\theta)$$

> subs(P=k*(theta-beta)/L/sin(theta),k-P*L*cos(theta));

$$k - \frac{k(\theta - \beta) \cos(\theta)}{\sin(\theta)}$$

> solve(k-k*(theta-beta)/sin(theta)*cos(theta)>0);

$$\left\{ k = k, \beta < \frac{\cos(\theta) \theta - \sin(\theta)}{\cos(\theta)} \right\}, \left\{ k = k, \frac{\cos(\theta) \theta - \sin(\theta)}{\cos(\theta)} < \beta \right\}$$

Para $\theta > 0$, simplificando, temos que: $\tan(\theta) > \theta - \beta$

Para imperfeições iniciais β pequenas, tem-se:

> plot([(theta-1)/sin(theta),(theta-2)/sin(theta),1+1/2*theta^2-(1-cos(theta))], theta=-2..2, lambda=-20..20, color = [red, blue, aquamarine]);




Figura 6 – Caminhos de Equilíbrio

Estruturas esbeltas, principalmente, comumente apresentam imperfeição inicial, como uma deformação angular. Uma estrutura real segue o caminho fundamental não linear indicado na parte positiva do eixo θ da Figura 6, o que indica que, para pequenos incrementos de carga, praticamente não se nota a rotação da coluna. Entretanto, ao se aproximar da carga crítica, a rotação do sistema passa a crescer, até que apareçam grandes rotações, o que na prática significa problemas estruturais, como fissuras ou plastificação. Ainda, percebe-se que em um sistema imperfeito não há bifurcação.

4. Barra rígida com mola linear lateral




Figura 7 – Posições Inicial e Final

No MAPLE:

> **U:=theta->1/2*k*(L*sin(theta))^2;**

$$U := \theta \rightarrow \frac{1}{2} k L^2 \sin(\theta)^2$$

> **We:=theta->P*L*(1-cos(theta));**

$$We := \theta \rightarrow P L (1 - \cos(\theta))$$

> **var_pot:=theta-> U(theta)-We(theta);**

$$var_pot := \theta \rightarrow U(\theta) - We(\theta)$$

> **var_pot(theta);**

$$\frac{1}{2} k L^2 \sin(\theta)^2 - P L (1 - \cos(\theta))$$

> **D(var_pot)(theta)=0;**

$$k L^2 \sin(\theta) \cos(\theta) - P L \sin(\theta) = 0$$

1ª Posição de Equilíbrio: (objeto não deformado) - caminho fundamental

> **theta=0;**

$$\theta = 0$$

2ª Posição de Equilíbrio: (objeto deslocado devido a carga P) - caminho pós crítico

> **solve(D(var_pot)(theta)=0,P);**

$$L \cos(\theta) k$$

Na 2ª Posição, para pequenas deflexões, o problema pode ser linearizado, de forma que:

> **solve([P=L*cos(theta)*k,cos(theta)=1]);** encontra-se a carga crítica $P_{cr} = k * L$
 $\{ L = L, P = L k, k = k, \theta = 0 \}$

Estabilidade dos caminhos: estudar a segunda derivada da energia potencial.

> **D(D(var_pot))(theta);**
 $k L^2 \cos(\theta)^2 - k L^2 \sin(\theta)^2 - P L \cos(\theta)$

Caminho Fundamental:

Como visto anteriormente, o equilíbrio no caminho fundamental se dá em $\theta = 0$, então pode-se simplificar a expressão da segunda derivada de forma que:

> **D_2:=k*L^2-P*L;**
 $D_2 := L^2 k - L P$
 > **solve([D_2>0,L>0,k>0,cos(theta)>0],P);**
 $\left\{ \begin{array}{ll} [\{ P < L k \}] & \text{And}(0 < L, 0 < k) \\ [] & \text{otherwise} \end{array} \right.$

Segunda derivada apresenta valores positivos quando $P < P_{cr}$. Assim, quando $P < P_{cr}$, tem-se equilíbrio **estável**.

> **solve([D_2<0,L>0,k>0,cos(theta)>0],P);**
 $\left\{ \begin{array}{ll} [\{ L k < P \}] & \text{And}(0 < L, 0 < k) \\ [] & \text{otherwise} \end{array} \right.$

De mesmo modo, quando $P > P_{cr}$, tem-se equilíbrio **instável**.

> **solve([D_2=0,L>0,k>0,cos(theta)>0],P);**
 $\left\{ \begin{array}{ll} [\{ P = L k \}] & \text{And}(0 < L, 0 < k) \\ [] & \text{otherwise} \end{array} \right.$

Quando a segunda derivada é igual a zero, $P = P_{cr}$, o equilíbrio é **indeterminado, inconclusivo**, o que não é suficiente para determinar a estabilidade do caminho. Deve-se analisar derivadas de ordem superior. ²

Caminho Pós-Crítico:

> **subs(P=k*L*cos(theta),D(D(var_pot))(theta));**
 $-k L^2 \sin(\theta)^2$

² Para analisar derivadas de ordem superior, podemos fazer a expansão em série de Taylor de $\text{var_pot}(\theta=0)$, e determinar a primeira derivada não nula (porque enquanto derivada = 0, não podemos concluir nada; na primeira não nula, estudamos quando é positiva ou negativa, para então concluirmos sobre a estabilidade deste ponto de bifurcação).

> solve([-k*L^2*sin(theta)^2<0,k>0,L>0]);

{ k = k, θ = 0, 0 < L }

A segunda derivada é negativa para todo e qualquer valor de θ , logo o caminho pós crítico é sempre **instável**.

> taylor(var_pot(theta),theta=0);

$$\left(\frac{1}{2}kL^2 - \frac{1}{2}LP\right)\theta^2 + \left(-\frac{1}{6}kL^2 + \frac{1}{24}LP\right)\theta^4 + O(\theta^6)$$

Sendo $k = P/L > 0$, o primeiro termo não nulo é $-\frac{1}{6}kL^2 + \frac{1}{24}LP$.

> subs(P=k*L,-1/6*k*L^2+1/24*L*P);

$$-\frac{kL^2}{8}$$

Esse valor é sempre *negativo*, sendo assim, em $P = P_{cr}$ para $\theta = 0$, a configuração é **instável**.

> with(plots):

> implicitplot([lambda=cos(theta),theta=0],theta=Pi/2..Pi/2,lambda=0..1,color=[red,blue], thickness=[2,3]);




Figura 8 – Caminhos de Equilíbrio

A estrutura, ao ser carregada inicialmente, segue pelo caminho fundamental representado na Figura 8, pré-crítico indicado pela cor azul, que é estável até o ponto de bifurcação, onde $\lambda = 1$, logo, $P = P_{cr}$. A partir deste ponto, quando a carga ultrapassa o valor crítico, a estrutura passa a ser instável (curva vermelha), com rigidez negativa.

```
>plot([1/2*sin(theta)^2-1/2*(1-cos(theta)),1/2*sin(theta)^2-0.7*(1-cos(theta)),
1/2*sin(theta)^2-(1-cos(theta)), 1/2*sin(theta)^2-2*(1-cos(theta))], theta=-1.6..1.6,
var_pot=-0.1..0.3, color=[red,green,gold,blue]);
```




Figura 9 – Variação da Energia Potencial

A Figura 9 representa as variações da energia potencial. Até $P = P_{cr}$, a estrutura apresenta apenas um mínimo, sendo este associado ao equilíbrio estável. A partir de $P \geq P_{cr}$, o gráfico da variação de energia potencial apresenta apenas um máximo, na origem, que representa um equilíbrio instável. Como analogia a uma situação física, nos gráficos vermelho e verde (ambos $P < P_{cr}$), um objeto introduzido ali sempre se manteria na mesma posição. Porém, ao aumentar a carga, de modo que fique maior ou igual a carga crítica ($P \geq P_{cr}$, gráficos marrom e azul), o objeto não teria como ficar estável, sendo obrigado a cair para algum dos lados em busca da estabilidade, que nunca será atingida. Sendo assim, considera-se que houve uma diminuição da região segura da estrutura. Uma estrutura como uma casca, por exemplo, se manteria na vertical até pouco antes de $P = P_{cr}$, mas depois, se $P \geq P_{cr}$, passaria a sofrer alguma deformação. Para $P < P_{cr}$ tem-se dois máximos relativos às posições instáveis (em vermelho). Se a estrutura ultrapassa estes picos ela se torna instável mesmo com $P < P_{cr}$. Diz-se que neste caso a estrutura apresenta uma bifurcação simétrica instável e a região segura é delimitada pelos dois picos.

5. Sistema com imperfeição inicial β




Figura 10 - Posições Inicial e Final

As imperfeições iniciais, por sua vez, podem ser de origem geométrica, ou devido às propriedades mecânicas variáveis do material, à excentricidade de carregamento, à carregamentos acidentais laterais, dentre outros [2]. Para a estrutura imperfeita tem-se:

> **U:=theta->1/2*k*L^2*(sin(theta)-sin(beta))^2;**

$$U := \theta \rightarrow \frac{1}{2} k L^2 (\sin(\theta) - \sin(\beta))^2$$

> **We:=theta->P*L*(cos(beta)-cos(theta));**

$$We := \theta \rightarrow P L (\cos(\beta) - \cos(\theta))$$

> **var_pot:=theta-> U(theta)-We(theta);**

$$var_pot := \theta \rightarrow U(\theta) - We(\theta)$$

> **var_pot(theta);**

$$\frac{1}{2} k L^2 (\sin(\theta) - \sin(\beta))^2 - P L (\cos(\beta) - \cos(\theta))$$

> **D(var_pot)(theta)=0;**

$$k L^2 (\sin(\theta) - \sin(\beta)) \cos(\theta) - P L \sin(\theta) = 0$$

> **solve(%);**

$$\begin{aligned} & \{ L = 0, P = P, \beta = \beta, k = k, \theta = 0 \}, \{ L = L, P = P, \beta = \beta, k = 0, \theta = 0 \}, \\ & \{ L = L, P = P, \beta = \beta, k = 0, \theta = \pi \}, \{ L = L, P = P, \beta = 0, k = k, \theta = 0 \}, \\ & \{ L = L, P = P, \beta = 0, k = k, \theta = \pi \}, \\ & \{ L = L, P = \frac{(\sin(\theta) - \sin(\beta)) L k \cos(\theta)}{\sin(\theta)}, \beta = \beta, k = k, \theta = \theta \} \end{aligned}$$

Sabendo que $\lambda = P / (k * L)$ e simplificando a expressão acima tem-se:

> **lambda:=theta->cos(theta)*(1-sin(beta)/sin(theta));**

P:=theta->k*L*cos(theta)*(1-sin(beta)/sin(theta));

$$\lambda := \theta \rightarrow \cos(\theta) \left(1 - \frac{\sin(\beta)}{\sin(\theta)} \right)$$

$$P := \theta \rightarrow k L \cos(\theta) \left(1 - \frac{\sin(\beta)}{\sin(\theta)} \right)$$

> **D(D(var_pot));**

$$\theta \rightarrow k L^2 \cos(\theta)^2 - k L^2 (\sin(\theta) - \sin(\beta)) \sin(\theta) - P L \cos(\theta)$$

> **simplify(subs(P=-(-sin(theta)+sin(beta))*L*k*cos(theta)/sin(theta),k*L^2*cos(theta)^2-k*L^2*(sin(theta)-sin(beta))*sin(theta)-P*L*cos(theta)));**

$$\frac{k L^2 (\cos(\theta)^2 \sin(\theta) - \sin(\theta) + \sin(\beta))}{\sin(\theta)}$$

Colocando $\sin(\theta)$ em evidência no numerador, tem-se $\cos(\theta)^2 - 1$, que é $\sin(\theta)^2$, encontrando-se então que:

> **D_2:=k*L^2*(-sin(theta)^3+sin(beta))/sin(theta);**

$$D_2 := \frac{k L^2 (-\sin(\theta)^3 + \sin(\beta))}{\sin(\theta)}$$

Quando a carga máxima (ou ponto limite) que P pode assumir ocorre, o ângulo de deformação da barra também atinge o seu limite.

Para estudar a estabilidade do equilíbrio:

> **solve((-sin(theta)^3+sin(theta[0]))/sin(theta),theta);**

$$\arcsin(\sin(\theta_0)^{(1/3)}), \arcsin\left(-\frac{1}{2}\sin(\theta_0)^{(1/3)} + \frac{1}{2}I\sqrt{3}\sin(\theta_0)^{(1/3)}\right), \\ -\arcsin\left(\frac{1}{2}\sin(\theta_0)^{(1/3)} + \frac{1}{2}I\sqrt{3}\sin(\theta_0)^{(1/3)}\right)$$

Tem-se que $\theta_{\text{lim}} = \arcsin(\sin(\theta_0)^{(1/3)})$

> **solve((-sin(theta)^3+sin(1))/sin(theta)<0,theta);**

$$\text{RealRange}(\text{Open}(-4\pi + \arcsin(\sin(1)^{(1/3)})), \text{Open}(-\arcsin(\sin(1)^{(1/3)}) - 3\pi)), \\ \text{RealRange}(\text{Open}(-\arcsin(\sin(1)^{(1/3)}) - 3\pi), \text{Open}(-2\pi)), \\ \text{RealRange}(\text{Open}(-2\pi + \arcsin(\sin(1)^{(1/3)})), \text{Open}(-\arcsin(\sin(1)^{(1/3)}) - \pi)), \\ \text{RealRange}(\text{Open}(-\pi), \text{Open}(0)), \\ \text{RealRange}(\text{Open}(\arcsin(\sin(1)^{(1/3)})), \text{Open}(-\arcsin(\sin(1)^{(1/3)}) + \pi)), \\ \text{RealRange}(\text{Open}(\pi), \text{Open}(2\pi)), \\ \text{RealRange}(\text{Open}(2\pi + \arcsin(\sin(1)^{(1/3)})), \text{Open}(-\arcsin(\sin(1)^{(1/3)}) + 3\pi)), \\ \text{RealRange}(\text{Open}(-\arcsin(\sin(1)^{(1/3)}) + 3\pi), \text{Open}(4\pi + \arcsin(\sin(1)^{(1/3)}))), \\ \text{RealRange}(\text{Open}(4\pi + \arcsin(\sin(1)^{(1/3)})), \text{Open}(-\arcsin(\sin(1)^{(1/3)}) + 5\pi))$$

$D_2 < 0$ quando $\theta_{\text{lim}} < \theta$, ou seja, quando o ângulo ultrapassa o ângulo limite, a configuração do modelo é **instável**.

Por conseguinte:

$D_2 > 0$ quando $\theta_{\text{lim}} > \theta$, ou seja, até o ângulo ultrapassar o ângulo limite, a configuração do modelo é **estável**.

> **plot([cos(theta)*(1-sin(0.01))/sin(theta)),cos(theta)], theta=-Pi/2..Pi/2, lambda=0..1.2, legend=["modelo imperfeito", "modelo s/ deformação inicial"], thickness=2, color=[red,green]);**




Figura 11 – Caminhos de Equilíbrio

A curva vermelha da Figura 11 representa a resposta a carregamentos da estrutura com imperfeição inicial, sendo o caminho da esquerda o caminho complementar, que é instável, e o caminho da direita o caminho principal. Quando a estrutura sofre pequenas deformações, ela mantém sua estabilidade, mas ao atingir o ângulo limite (ponto máximo da função, no eixo θ positivo) perde essa estabilidade e passa a ser instável.

A curva verde foi demonstrada anteriormente no Item 4 Figura 8, e representa a configuração de um modelo estrutural sem imperfeição inicial, sendo completamente instável.

```
> plot([cos(theta)*(1-sin(0)/sin(theta)), cos(theta)*(1-sin(0.5)/sin(theta)), cos(theta)*(1-sin(0.1)/sin(theta)), cos(theta)*(1-sin(0.2)/sin(theta)), cos(theta)*(1-sin(0.3)/sin(theta))], theta=-1..1, Pmax=0..1.2, legend=["tetha0=0", "tetha0=0.5", "tetha0=0.1", "tetha0=0.2", "tetha0=0.3"], color=[black,red,blue,green,gold]);
```




Figura 12 – Comportamento de estruturas com e sem imperfeições iniciais

Na Figura 12, a curva preta representa um modelo estrutural sem imperfeições iniciais, diferentemente das demais. Fica claro que o comportamento estrutural é muito diferente entre elas: as imperfeições tendem a diminuir a capacidade de carga das estruturas. Por isso, as normas internacionais tendem a incorporar o efeito das imperfeições na análise de elementos estruturais esbeltos, o que faz com que a carga crítica máxima aceitável diminua, para garantir a segurança do projeto estrutural.

6. Flambagem Assimétrica




Figura 13 – Posições Inicial e Final

No MAPLE:

> **U:=theta->k*(L*(sqrt(1+sin(theta))-1))^2;**

$U := \theta \rightarrow k L^2 (\sqrt{1 + \sin(\theta)} - 1)^2$

> **We:=theta->P*L*(1-cos(theta));**

$We := \theta \rightarrow P L (1 - \cos(\theta))$

> **var_pot:=theta->U(theta)-We(theta);**

$var_pot := \theta \rightarrow U(\theta) - We(\theta)$

> **var_pot(theta);**

$k L^2 (\sqrt{1 + \sin(\theta)} - 1)^2 - P L (1 - \cos(\theta))$

> **D(var_pot)(theta)=0;**

$\frac{k L^2 (\sqrt{1 + \sin(\theta)} - 1) \cos(\theta)}{\sqrt{1 + \sin(\theta)}} - P L \sin(\theta) = 0$

> solve(k*L^2*(sqrt(1+sin(theta))-1)*cos(theta)/sqrt(1+sin(theta))-P*L*sin(theta) = 0,P);

$$\frac{k L (\sqrt{1 + \sin(\theta)} - 1) \cos(\theta)}{\sin(\theta) \sqrt{1 + \sin(\theta)}}$$

> lambda:=theta->(sqrt(1+sin(theta))-1)*cos(theta)/(sin(theta)*sqrt(1+sin(theta)));

$$\lambda := \theta \rightarrow \frac{(\sqrt{1 + \sin(\theta)} - 1) \cos(\theta)}{\sin(\theta) \sqrt{1 + \sin(\theta)}}$$

Linearizando essa solução, ou seja, $\sin(\theta) = \theta$ e $\cos(\theta) = 1$, tem-se que $P_{cr} = k*L/2$.

> subs(P=k*L/2,k*L^2*(2+2*cos(theta))^(1/2)-P*L*(1-sin(theta)));

$$k L^2 \sqrt{2 + 2 \cos(\theta)} - \frac{1}{2} k L^2 (1 - \sin(\theta))$$

> cos(theta)*(sqrt(2+2*sin(theta))-sqrt(2))/(sqrt(2+2*sin(theta))*sin(theta));

$$\frac{\cos(\theta) (\sqrt{2 + 2 \sin(\theta)} - \sqrt{2})}{\sqrt{2 + 2 \sin(\theta)} \sin(\theta)}$$

> plot((lambda(theta)),theta=-Pi/2..Pi/2);




Figura 14 – Caminho de Equilíbrio

Nota-se, na Figura 14, que o caminho pós-crítico possui uma inclinação inicial diferente de zero, o que caracteriza uma bifurcação assimétrica. Portanto, dependendo do sinal da imperfeição, pode-se ter dois comportamentos distintos. Para imperfeições positivas tem-se um comportamento similar ao das estruturas com bifurcação instável, com um ponto limite, sendo a capacidade de carga dada pela carga limite. Para imperfeições negativas tem-se um comportamento similar ao das estruturas com bifurcação instável, sendo o caminho de equilíbrio da estrutura imperfeita sempre estável.

7. Estrutura não linear abatida com bifurcação por ponto limite




Figura 15 – Posições Inicial e Final

No MAPLE:

> **U:=theta->1/2*k*(2*L)^2*((cos(theta)-cos(beta))^2);**

$$U := \theta \rightarrow 2 k L^2 (\cos(\theta) - \cos(\beta))^2$$

> **We:=theta->P*L*(sin(beta)-sin(theta));**

$$We := \theta \rightarrow P L (\sin(\beta) - \sin(\theta))$$

> **var_pot:=theta-> U(theta)-We(theta);**

$$var_pot := \theta \rightarrow U(\theta) - We(\theta)$$

> **var_pot(theta);**

$$2 k L^2 (\cos(\theta) - \cos(\beta))^2 - P L (\sin(\beta) - \sin(\theta))$$

> **D(var_pot)(theta)=0;**

$$-4 k L^2 (\cos(\theta) - \cos(\beta)) \sin(\theta) + P L \cos(\theta) = 0$$

> **solve(%,P);**

$$\frac{4 k L (\cos(\theta) - \cos(\beta)) \sin(\theta)}{\cos(\theta)}$$

Simplificando, tem-se, para imperfeições iniciais muito pequenas:

> **P:=theta->4*sin(theta)*(1-cos(1/180)/cos(theta));**

$$P := \theta \rightarrow 4 \sin(\theta) \left(1 - \frac{\cos\left(\frac{1}{180}\right)}{\cos(\theta)} \right)$$

> **plot(P(theta),theta=-1/180..1/180);**




Figura 16 – Carga Crítica

> **D(D(var_pot))(theta);**

$$4 k L^2 \sin(\theta)^2 - 4 k L^2 (\cos(\theta) - \cos(\beta)) \cos(\theta) - P L \sin(\theta)$$

Simplificando, tem-se que:

> **D_2:=4*k*L^2*(sin(theta)^2-cos(theta)^2+cos(beta)*cos(theta))-P*L*sin(theta);**

$$D_2 := 4 k L^2 (\sin(\theta)^2 - \cos(\theta)^2 + \cos(\beta) \cos(\theta)) - P L \sin(\theta)$$

>**plot(4*(sin(theta)^2-2*cos(theta)^2+cos(1/180)*cos(theta))-1/3*sin(theta),theta=-0.75..0.8);**




Figura 17 – Caminho de Equilíbrio

Conclusões:

O estudo teórico e o desenvolvimento do presente trabalho possibilitaram uma boa compreensão inicial acerca do tema de instabilidade estrutural. Foi possível entender a necessidade da determinação de cargas críticas de projeto de forma prática, visto que para cargas superiores, a estrutura se deforma, apresentando esforços e deslocamentos não presentes no projeto, podendo ocasionar plastificação ou fissuração e até mesmo o colapso da estrutura. Essa deformação é responsável pela diminuição da região segura das estruturas, o que afeta sua rigidez e logo, segurança.

Ainda, foi observada também a diferença entre os comportamentos de estruturas com e sem imperfeições iniciais. Por este motivo, as normas internacionais tendem a incorporar o efeito das imperfeições na análise da capacidade de carga de elementos estruturais esbeltos, o que faz com que a carga crítica máxima aceitável diminua.

Referências:

1. GONÇALVES, Paulo B. **Aula 2: Bifurcação Simétrica Estável**. 201-?. 34 slides.
2. GONÇALVES, Paulo B. **Aula 3 – Bifurcação Simétrica Instável**. 201-?. 29 slides.
3. TELES, Bruno de O., JÚNIOR, Paulo H. D., QUINAN, Túlio de B. **Avaliação da estabilidade global em estruturas em aço**. Universidade Federal de Goiás, 2016. Disponível em: https://files.cercomp.ufg.br/weby/up/140/o/AVALIA%C3%87%C3%83O_DA_ESTABILIDADE_GLOBAL_EM ESTRUTURAS EM A%C3%87O.pdf. Acesso em: 17 jun. 2020.